
 MODEL:
SETS:
  PRODUCT / ROCKET, METEOR, STREAK, COMET, JET, TURTLE /
                            : PROFIT, QTY, SETUP, BUILD ;
  RESOURCES / STEEL, COPPER, PLASTIC, RUBBER, GLASS, PAINT /
                             : AVAIL ;
  RXP( RESOURCES, PRODUCT): USAGE;
ENDSETS

[MAX] MAX = @SUM( PRODUCT( I):
   PROFIT( I) * QTY( I) - SETUP( I) * BUILD( I));
@FOR( RESOURCES( I):
   @SUM( PRODUCT( J): USAGE( I, J) * QTY( J)) <= AVAIL( I));

@FOR( PRODUCT( I):
   QTY( I) <= 1000 * BUILD( I);
   @BIN( BUILD( I));
   @GIN( QTY( I)));

DATA:
   !           ROCKET  METEOR  STREAK  COMET   JET  TURTLE;

   PROFIT =        30      45      24     26    24      30;
   SETUP =         35      20      60     70    75      30;

   AVAIL = !STEEL;  800
          !COPPER; 1160
         !PLASTIC; 1780
          !RUBBER; 1050
           !GLASS; 1360
           !PAINT; 1240;

   USAGE = !STEEL;  1       4       0      4     2       0
          !COPPER;  4       5       3      0     1       0
         !PLASTIC;  0       3       8      0     1       0
          !RUBBER;  2       0       1      2     1       5
           !GLASS;  2       4       2      2     2       4
           !PAINT;  1       4       1      4     3       4;

ENDDATA
 END


